

趣味



计算机程序设计科普读物
Visual FoxPro程序设计教学配套教参
计算机等级考试与程序设计竞赛参考书

Visual FoxPro 程序设计

杨克昌 刘志辉 编 著



君上程序设计



本书特色

- 突出求解问题生动有趣
- 注重求解程序简单易懂
- 力求问题结果直观明了
- 重视程序变通与问题引申



中国水利水电出版社
www.waterpub.com.cn

趣味 Visual FoxPro 程序设计集锦

杨克昌 刘志辉 编著



中国水利水电出版社
www.waterpub.com.cn

内 容 提 要

本书以各类趣题的 Visual FoxPro (简称 VFP) 程序设计求解为主线, 取材注重典型性与趣味性, 程序设计注重结构化与可读性。精选的趣解包括典型的数值求解, 常见的数据处理, 有趣的智力游戏, 巧妙的模拟探索, 新颖的图表创建。大多趣题是引导入门的基础题、常规题, 也适当涉及少量难度较大的综合题与经典名题, 难度适宜, 深入浅出。

为适应不同基础的读者学习与欣赏, 对有些趣题采用多种思路设计与多个程序实现。其中少量难度较大、要求较高的问题, 在目录中用“*”标注, 可供学习“VFP 程序设计”课程的同学进行课程设计时选用。

本书着眼于应用 VFP 程序设计求解问题的基本方法与技巧, 可作为 VFP 程序设计的科普读物与学习 VFP 程序设计的教学参考书, 以期提高通过 VFP 程序设计解决实际问题的能力。

本书适合普通高校本专科学学生、职业技术学院学生与程序设计爱好者学习 VFP 程序设计时参考, 也可供参加各级程序设计选拔赛、计算机等级考试与计算机程序员水平考试参考, 还可供中学信息学(计算机)奥林匹克指导与 IOI, NOI 培训选用。

图书在版编目(CIP)数据

趣味 Visual FoxPro 程序设计集锦 / 杨克昌, 刘志辉
编著. -- 北京: 中国水利水电出版社, 2010.1
ISBN 978-7-5084-6985-0

I. ①趣… II. ①杨… ②刘… III. ①关系数据库—
数据库管理系统, Visual FoxPro—程序设计 IV.
①TP311.138

中国版本图书馆CIP数据核字(2009)第209132号

策划编辑: 杨庆川

责任编辑: 李 炎

封面设计: 李 佳

书 名	趣味 Visual FoxPro 程序设计集锦
作 者	杨克昌 刘志辉 编著
出版发行	中国水利水电出版社 (北京市海淀区玉渊潭南路1号D座 100038) 网址: www.waterpub.com.cn E-mail: mchannel@263.net (万水) sales@waterpub.com.cn 电话: (010) 68367658 (营销中心)、82562819 (万水)
经 售	全国各地新华书店和相关出版物销售网点
排 版	北京万水电子信息有限公司
印 刷	北京市天竺颖华印刷厂
规 格	170mm×237mm 16开本 22印张 451千字
版 次	2010年1月第1版 2010年1月第1次印刷
印 数	0001—3000册
定 价	35.00元

凡购买我社图书, 如有缺页、倒页、脱页的, 本社营销中心负责调换
版权所有·侵权必究

前 言

从当前高校各专业计算机语言课程的系统开设,到中小学信息技术(计算机)课的相继开出,推动了计算机文化在神州大地的广泛普及。为顺应信息技术迅猛发展与计算机教育不断深入的潮流,帮助包括大中专在校学生在内的广大青少年逐步掌握计算机程序设计的设计思路与基本技能,在程序设计中开拓求解思路,解决实际问题,培养创新意识,激发提出问题与应用程序设计解决问题的欲望与兴趣,不断提高程序设计水平与应用求解能力,是我们计算机教育工作者义不容辞的职责。

当前,大专院校理工科专业一般学习 C (C++) 语言程序设计,而文科专业一般学习 VFP 程序设计。各种面向大中专院校的程序设计教材繁杂而重复,而配合程序设计教学的参考书与课外读物却是凤毛麟角。鉴于此,特推出计算机程序设计科普读物《趣味 C 程序设计集锦》与《趣味 Visual FoxPro 程序设计集锦》,以期推动广大本专科学生与程序设计爱好者学习程序设计的不断深入,促进程序设计水平与应用求解能力的逐步提高。

学习计算机语言的目的是什么?当然是程序设计!那么,程序设计的目的又是什么?毫无疑问,程序设计的目的是求解问题,求解靠人工计算或推理一时无从下手的各类实际应用问题。你应用程序设计解决的问题越多、越普遍、越深刻,你的程序设计成绩就越大,你的程序设计能力就越强。而应用程序设计求解各类实际应用问题,恰恰是现今大多数程序设计教材所忽略的环节,也是造成广大在校学生学习程序设计语言缺乏兴趣的直接原因。

本书以各类中外趣味问题的程序设计求解为主线,取材注重趣味性与典型性,题型丰富多彩,内容新颖丰富。精选程序设计趣味问题,包括各类整数求解,数据处理,智力游戏,模拟探索,图表创建,大多是引导入门的基础题与常规题,也适当涉及少量难度较大的综合题与经典名题,难度适宜,深入浅出。书中部分选题取自国际国内信息学(计算机)奥林匹克与各类程序设计竞赛,同时参考了网上读者集中探讨的程序设计热点问题,有利于高校学生与程序设计爱好者在计算机实例求解上开阔视野,在程序设计思路开拓与应用技巧上有一个深层次的练习与提高。部分难度较大、要求较高的问题在目录中用“*”标注,可供在校学生进行课程设计时选用。

本书力求突出以下四个特色:

突出求解问题生动有趣。选取的程序设计趣题应为程序设计爱好者所喜爱。这些趣题通常是著名的中外名题,或应用常规的推理与人工计算难以解决的问题,以充分体现程序设计求解的优势。为了避免一味拔高而降低趣味性,删除了若干枯燥难懂、学术性较强的问题。

注重求解程序简单易懂。题解程序选用广大青少年与在校本专科学生正在学习的、

使用率最高的计算机语言编写程序。同时在算法上尽量避免使用学术性强的专业算法。对其中有些典型问题，采用不同设计思路与表现形式设计出不同的求解程序，以适应程序设计基础不同的读者学习与欣赏。

力求问题**结果直观明了**。尽可能给出求解程序的输出结果与运行示例，并作必要的讨论与分析，使读者对问题求解结果一目了然，以帮助读者对所求解问题的清晰理解与对设计程序的深入掌握。

重视**程序变通与问题引伸**。对求解程序进行必要的改进、变通与优化，是促进应用程序设计实际问题能力的培养，切实提高程序设计水平的必要手段与有效途径。同时，对有些求解趣题作适当的引伸与拓广，引导有兴趣的爱好者对相关问题作进一步的探索与研究。

本书作为计算机程序设计的教学参考书与科普读物，适合在校本专科学生与广大程序设计爱好者学习参考，也可供各级程序设计选拔赛与国际大学生程序设计竞赛（ACM）、计算机等级考试与计算机程序员水平考试复习使用，也可供中学信息学（计算机）奥林匹克指导与 IOI, NOI 及各省程序设计竞赛培训参考。

在书稿的编写过程中，湖南理工学院计算机学院院长王岳斌教授、周持中教授以及严权锋、郭华等老师提出了很好的修改意见，笔者在此深表感谢！

尽管每一道题解都经反复检查，每一个程序都经多次运行调试，因涉及内容较广，难免存在各种差错，恳请读者指正。

杨克昌 刘志辉

2009 年 10 月

目 录

前言

一、奥运奖牌榜——数据表的处理范例	1
1 数据表记录、字段顺序调整	1
2 含个人所得税的工资表处理	4
3 奥运奖牌榜	6
4 大奖赛现场统分	9
二、舍罕王的失算——不可忽视的和与积	12
5 舍罕王的失算	12
6 分数不等式	14
7 阶乘与阶乘和数	16
8 综合高精度计算	19
9 图形点扫描	22
三、勾股数——古老文明的精华	25
10 最大公约数与最小公倍数	25
11 水仙花数	28
12 勾股数	31
13 完全数	35
14 相亲数	39
15 守形数	41
四、素数——上帝用来描写宇宙的文字	45
16 素数	45
17 乌兰现象	48
18 孪生素数	51
19 梅森尼数	53
20 金蝉素数	55
21 素数多项式	57
22 等差素数列	60
23 验证歌德巴赫猜想	62
24 合数世纪探求	63
五、桥本分数式——优美的智慧	67
25 逆序乘积式	67

26	巧妙的三组平方	71
27	完美和式	74
28	完美乘积式	77
29	完美综合运算式	80
30	桥本分数式	83
31	埃及分数式	87
六、斐波那契序列——递推的学问		92
32	分数数列	92
33	斐波那契序列与卢卡斯序列	95
34	幂序列	99
35	双关系递推数列	103
36	基于 $2x+3y$ 的递推数列	105
37	汉诺塔问题	107
38	猴子吃桃	110
39	猴子爬山	112
40	购票排队	115
*41	神秘的数组	116
七、韩信点兵——远古的神机妙算		123
42	破解数字魔术	123
43	鸡兔同笼与羊犬鸡兔问题	124
44	百鸡问题	127
45	韩信点兵	129
46	整币兑零	131
*47	解佩尔方程	134
八、泊松分酒——奇妙的分解		140
48	分解质因数	140
49	积最大的整数分解	143
50	整数的分划	145
*51	泊松分酒	148
*52	西瓜分堆	153
*53	水手分椰子	160
九、角谷猜想——精巧的转化		166
54	分数化小数	166
55	金额大写转化	168
56	数制转换	169

57 角谷猜想	172
58 黑洞数 495 与 6174	174
59 回文数	178
十、幻方——古今中外的数阵奇葩	183
60 杨辉三角	183
61 数字三角形	185
62 折叠方阵与旋转方阵	190
*63 幻方	195
*64 三阶素数幻方	202
十一、插入乘号——决策的最优化	206
65 数列最优压缩	206
*66 最长公共子串与子序列	207
67 删除中的最值问题	212
68 古尺神奇	215
69 数码珠串	219
*70 数阵中的最优路径	223
*71 插入乘号问题	228
72 智能甲虫的安全点	233
73 点的覆盖圆	236
十二、尾数前移——运算模拟的典范	239
74 均位奇观探索	239
75 多少个 1 能被 2009 整除	241
76 01 串积问题	244
77 连写数整除问题	247
78 尾数前移问题	251
*79 求圆周率 π 到 n 位	253
十三、外索夫游戏——博弈策略的秘诀	257
80 围圈循环报数	257
*81 围圈中的无忧位与绝望位	258
82 列队顺逆报数	262
83 洗牌复原	264
84 翻币倒面	267
85 黑白棋子移动	269
86 外索夫游戏	271
十四、多格式万年历——变幻多姿的图表	276

87 新颖的 p 进制乘法表	276
88 多格式万年历	280
89 金字塔图案	283
90 带中空金字塔	286
91 菱形与灯笼图案	289
92 函数 $y=\sin(x)/x$ 图形	297
十五、高斯八皇后——排列组合的精彩	299
93 排列中的平方数	299
94 实现 $A(n, m)$ 与若干复杂排列	302
95 实现 $C(n, m)$ 与允许重复组合	309
96 高斯八后问题	312
*97 皇后控制棋盘问题	316
*98 伯努利装错信封问题	323
*99 别出心裁的情侣拍照	327
*100 德布鲁金环序列	329
附录	335
附录 A VFP 语法提要	335
附录 B VFP 常用函数	338
参考文献	343

一、奥运奖牌榜——数据表的处理范例

1 数据表记录、字段顺序调整

1. 问题提出

VFP 操作命令与系统函数非常丰富，但没有任意调整表中记录物理顺序的命令，也没有简单调整字段顺序的函数。为操作方便，试设计交换数据表中两指定记录的位置、交换表中两指定字段位置的自定义函数。多次交换即可完成任意顺序的调整。

2. 交换数据表中两指定记录位置函数设计

(1) 设计思路

在 VFP 系统中可通过排序改变记录的物理顺序，也可通过索引改变记录的逻辑顺序。如果我们要交换已有数据表中两指定记录（例如第 m 条记录与第 n 条记录）的位置，需要定义数组，通过数据从当前记录传送到数组命令 `scatter` 与从数组覆盖当前记录的命令 `gather` 来完成交换。

(2) 自定义函数 (swr.prg) 设计

* 交换两指定记录位置 f011

```
function swr
parameters m,n
t=reccount()           && t 为表的记录数
if m>t or n>t
    ? [参数超界! ]
    return
endif
k=fcount()             && k 为表的字段数
dimension szm(k),szn(k) && 定义用于交换的两个数组
? "    原表为: "
list off
go m
scatter to szm memo     && 第 m 条记录到 szm 数组
go n
scatter to szn memo     && 第 n 条记录到 szn 数组
```

```

go m
gather from szn memo      &&  szn 数组覆盖当前记录
go n
gather from szm memo      &&  szm 数组覆盖当前记录
? "  交换指定记录位置后表为: "
list off                  &&  显示交换后的表记录
return

```

(3) 调用函数 swr() 交换记录示例

对已打开的“工资表”，在命令窗口输入命令：swr(2,4)

原表为：

姓名	基本工资	津贴	保险	所得税	实发
张山	3521.00	4213.00	165.40	771.80	6796.80
李大明	4831.00	3118.00	188.80	814.80	6945.40
杨义	4682.00	2131.00	276.50	596.95	5939.55
王天际	2361.00	4245.00	132.60	565.90	5907.50
刘小娟	3500.00	3230.00	156.00	584.50	5989.50

交换指定记录位置后表为：

姓名	基本工资	津贴	保险	所得税	实发
张山	3521.00	4213.00	165.40	771.80	6796.80
王天际	2361.00	4245.00	132.60	565.90	5907.50
杨义	4682.00	2131.00	276.50	596.95	5939.55
李大明	4831.00	3118.00	188.80	814.80	6945.40
刘小娟	3500.00	3230.00	156.00	584.50	5989.50

表中的第 2, 4 两条记录位置已完成交换，表中的所有数据没有改变。

3. 交换数据表中两指定字段位置函数设计

(1) 设计思路

要交换已打开的数据表中两指定字段（例如第 m 个字段与第 n 个字段）的位置，可应用数据表的复制命令，把交换后的字段顺序分段组织成一个字符串 fz，作为数据表复制命令中 field 子句的内容。

(2) 自定义函数 (swf.prg) 设计

* 交换两指定字段位置 f012

```

function swf
parameters m,n
if m>n                                &&  确保前一个参数小于后一个
    x=m
    m=n
    n=x
endif
k=fcount()                            &&  k 为当前表的字段数
if n>k

```

```

? [ 参数太大,超过表的字段数!]
return
endif
? " 原表为: "
list off
fz=[]                                && 以下分段组织字符串 fz
for j=1 to m-1
    fz=fz+field(j)+[,]
endfor
fz=fz+field(n)+[,]                  && 实施第 m,n 个字段互换
for j=m+1 to n-1
    fz=fz+field(j)+[,]
endfor
fz=fz+field(m)+[,]
for j=n+1 to k-1
    fz=fz+field(j)+[,]
endfor
fz=fz+field(k)
copy to newf field &fz              && 交换后的表为 newf.dbf
use newf
? " 交换指定字段位置后表为: "
list off
return

```

(3) 调用函数 swf() 交换字段示例

例如,对已打开的工资表,输入命令 swf(4,5),得
原表为:

姓名	基本工资	津贴	所得税	保险	实发
张山	3521.00	4213.00	0.00	165.40	0.00
王天际	2361.00	4245.00	0.00	132.60	0.00
杨义	4682.00	2131.00	0.00	276.50	0.00
李大明	4831.00	3118.00	0.00	188.80	0.00
刘小娟	3500.00	3230.00	0.00	156.00	0.00

交换指定字段位置后表为:

姓名	基本工资	津贴	保险	所得税	实发
张山	3521.00	4213.00	165.40	0.00	0.00
王天际	2361.00	4245.00	132.60	0.00	0.00
杨义	4682.00	2131.00	276.50	0.00	0.00
李大明	4831.00	3118.00	188.80	0.00	0.00
刘小娟	3500.00	3230.00	156.00	0.00	0.00

表中的第 4, 5 两个字段位置已完成交换,表中的所有数据没有改变。

2 含个人所得税的工资表处理

1. 问题提出

已有简单“工资表”如下，其中收入部分的“基本工资”与“津贴”两个字段有原始数据，需扣除部分的“保险”字段也有数据。需计算每人的“所得税”与“实发”。

姓名	基本工资	津贴	保险	所得税	实发
张山	3521.00	4213.00	165.40	0.00	0.00
王天际	2361.00	4245.00	132.60	0.00	0.00
林义	4682.00	2131.00	276.50	0.00	0.00
李文明	4831.00	3118.00	188.80	0.00	0.00
刘小娟	3500.00	3230.00	156.00	0.00	0.00

其中个人所得税的计算是工资表处理中的难点。

2. 个人所得税计算要点

个人所得税按九级超额累进税率进行计算。根据新的个人所得税计算规则，新的起征点为 $c=2000$ 元。收入扣除 2000 元后：

- 不超过 500 元的部分，征收 5%；
- 超 500~2000 元部分，征收 10%；
- 超 2000~5000 元部分，征收 15%；
- 超 5000~20000 元部分，征收 20%；
- 超 20000~40000 元部分，征收 25%；
- 超 40000~60000 元部分，征收 30%；
- 超 60000~80000 元部分，征收 35%；
- 超 80000~100000 元部分，征收 40%；
- 超 100000 元以上部分，征收 45%。

应用多分支情况语句 `do case—endcase` 计算所得税函数。设收入部分总和为 y ，个人所得税为 x 。为方便操作，设计求个人所得税自定义函数 `sds(y)`，返回所得税额 x 。

为简化设计，我们在多分支结构中应用了调用自身函数 `sds()`。

3. 个人所得税自定义函数设计

* 个人所得税计算自定义函数 (sds.prg) f021

```
function sds          && 函数名为 sds
parameters y
c=2000                && 确定起征点
do case
    case y<=c
```

```

x=0
case y<=c+500
  x=(y-c)*0.05
case y<=c+2000
  x=sds(c+500)+(y-c-500)*0.10    && 调用自身函数 sds()
case y<=c+5000
  x=sds(c+2000)+(y-c-2000)*0.15
case y<=c+2+20000
  x=sds(c+5000)+(y-c-5000)*0.20
case y<=c+40000
  x=sds(c+20000)+(y-c-20000)*0.25
case y<=c+60000
  x=sds(c+40000)+(y-c-40000)*0.30
case y<=c+80000
  x=sds(c+60000)+(y-c-60000)*0.35
case y<=c+100000
  x=sds(c+80000)+(y-c-80000)*0.40
othe
  x=sds(c+100000)+(y-c-100000)*0.45
endcase
return x

```

使用所得税函数 $sds(y)$ (其中括号内 y 为总收入数), 可简化对个人所得税的计算。例如, 某人收入为 15500 元, 在命令窗口输入: ? $sds(15500)$, 得个人所得税为: 2325.00

4. 含个人所得税工资表处理程序实现

* 工资表处理程序 f022

```

set talk off
use 工资表
replace all 所得税 with sds(基本工资+津贴)
replace all 实发 with 基本工资+津贴-保险-所得税
list off
return

```

运行程序, 结果如下:

姓名	基本工资	津贴	保险	所得税	实发
张山	3521.00	4213.00	185.40	771.80	6796.80
王天际	2361.00	4245.00	132.60	565.90	5907.50
杨义	4682.00	2131.00	276.50	596.95	5939.55
李大明	4831.00	3118.00	188.80	814.80	6945.40
刘小娟	3500.00	3230.00	156.00	584.50	5989.50

在自定义函数中设置起征点 $c=2000$, 为以后修改函数提供方便。例如, 若干时日, 起征点提高到 3000 元, 而其他税率未变, 则只需修改 $c=3000$ 即可。

3 奥运奖牌榜

1. 问题提出

奥运会、全运会等大型竞赛都会随竞赛进程及时公布经排序的“奖牌榜”。“奖牌榜”的排序规则通常为：

1) 首先按金牌数降序排列；当金牌数相同时按银牌数降序排列；当银牌数也相同时按铜牌数降序排列。

2) 当金银铜三种奖牌数全相同时，排名的名次相同。

已有“奖牌表”中设置有字段：名次(N)，单位(C)，金牌(N)，银牌(N)，铜牌(N)，总数(N)。其中“名次”与“总数”字段无数据，其他字段均为最新的数据。

试根据“奖牌表”上的金银铜三种奖牌的数据，依“奖牌榜”的排序规则产生“奖牌榜”。或在不改变“奖牌表”记录物理顺序前提下自动填写表中的“排名”数据。

2. 产生奖牌榜

(1) 设计要点

注意到奖牌榜改变了记录的物理顺序，可应用 VFP 的多级排序命令 sort：

```
sort to 奖牌榜 on 金牌/d, 银牌/d, 铜牌/d
```

为了实现当金银铜三种奖牌数全相同时，排名的名次相同，设置在按 sort 排序后的循环中每一条记录与上一记录比较：

若三种奖牌与上一记录不完全相同，则用本记录的记录号 recn() 替换“排名”名次。

若三种奖牌与上一记录完全相同，则用上一条记录的排名数据 m 替换“排名”名次。

(2) 产生奖牌榜程序设计

* 奖牌榜程序设计 f031

* 按金牌数由大到小排序，金牌数相同时按银牌数，再按铜牌数。

* 要求三种奖牌数相同时排名名次也相同。

```
use 奖牌表                                && 打开更新三种奖牌数据
list off                                    && 显示排序前的奖牌表
sort to 奖牌榜 on 金牌/d, 银牌/d, 铜牌/d    && 按奖牌榜要求排序
use 奖牌榜
stor 0 to a, b, c                            && a, b, c 赋初值
do while !eof()
    if 金牌#a or 银牌#b or 铜牌#c            && 三种奖牌与上一记录不全同
        m=recn()                             && 当前记录号赋值给 m
```

```

endif
replace 名次 with m           && 用 m 替换产生新的排名数据
a=金牌                       && 当前记录值赋值给 a, b, c
b=银牌
c=铜牌
skip                           && 记录指针下移
enddo
replace all 总数 with 金牌+银牌+铜牌   && 计算各记录奖牌总数
? [ 奖牌榜: ]
list off                       && 显示排序后的奖牌榜
return

```

(3) 运行程序示例

运行程序，显示如下：

名次	单位	金牌	银牌	铜牌	总数
0	A	15	14	21	0
0	B	12	13	12	0
0	C	17	15	11	0
0	D	5	6	7	0
0	E	12	13	12	0
0	F	11	14	10	0
0	G	5	6	7	0

奖牌榜：

名次	单位	金牌	银牌	铜牌	总数
1	C	17	15	11	43
2	A	15	14	21	50
3	B	12	13	12	37
3	E	12	13	12	37
5	F	11	14	10	35
6	D	5	6	7	18
6	G	5	6	7	18

3. 填写奖牌表中的名次数据

要求按奖牌榜的排序规则排序，但不改变奖牌表中各记录的物理顺序，在奖牌表中自动填写名次数据。

(1) 设计要点

注意到不能改变表中记录的物理顺序，显然不能用 sort 命令，只能用索引 index 命令来完成排序。

关键是索引表达式如何确定。索引表达式如果简单地定义为“金牌+银牌+铜牌”，势必按三种奖牌数的总和来排序，有违排序规定。

为突出主次三项排序的目的，注意到 VFP 中字符串比较是从左至右逐个字符按 ASCII 码比较，把三种奖牌数转换为字符串后相连作为索引表达式是适宜的。

例如设金牌数分别为 13、7，因“1”<“7”导致“13”<“7”；但用 str() 函数转换为字符串，str(13) 在“13”前有 8 个空格，而 str(7) 在“7”前有 9 个空格，相同位置的字符比较，显然有“1”>“ ”，因而导致 str(13)>str(7)。

因此，设置按以下命令索引：

```
index on str(金牌)+str(银牌)+str(铜牌) tag nf desc
```

可完成规定的排序。

(2) 填写奖牌表中的排名数据程序设计

* 要求排名不改变表记录的物理顺序 f032

* 首先按金牌数由高到低排序，相同时按银牌数，再按铜牌数。

* 三种奖牌数完全相同时名次相同。

```
set talk off
use 奖牌表                                && 打开并修改奖牌表数据
replace all 总数 with 金牌+银牌+铜牌      && 计算并替换各记录奖牌总数
index on str(金牌)+str(银牌)+str(铜牌) tag nf desc && 按排序要求索引
stor 0 to a,b,c                            && a,b,c 赋初值
go top
for k=1 to recc()                          && 按记录数循环
    if 金牌#a or 银牌#b or 铜牌#c          && 若三种奖牌数与上一记录不全相同
        m=k                                && 本记录逻辑顺序号赋值给 m
    endif
    replace 名次 with m                    && 用 m 替换填写本记录的名次
    a=金牌                                  && 按本记录的值为 a,b,c 赋值，为下次比较作准备
    b=银牌
    c=铜牌
    skip                                    && 记录指针下移
endfor
set index to                                && 关闭索引，恢复记录的原顺序
list off                                    && 显示填写了排名数据的奖牌表
return
```

(3) 程序运行示例

运行程序，得

名次	单位	金牌	银牌	铜牌	总数
2	A	15	14	21	50
3	B	12	13	12	37
1	C	17	15	11	43
6	D	5	6	7	18
3	E	12	13	12	37
5	F	11	14	10	35
6	G	5	6	7	18

可知奖牌表的各记录的物理顺序未变，已按排序要求自动填写了“名次”数据。

对其他各类竞赛或需排名次的活动，可参照本例完成排序。